﻿/*

  The contents of this file are subject to the Mozilla Public License Version
  1.1 (the "License"); you may not use this file except in compliance with
  the License. You may obtain a copy of the License at 
  
           http://www.mozilla.org/MPL/ 
  
  Software distributed under the License is distributed on an "AS IS" basis,
  WITHOUT WARRANTY OF ANY KIND, either express or implied. See the License
  for the specific language governing rights and limitations under the License. 
  
  The Original Code is AST'r Framework.
  
  The Initial Developer of the Original Code is
  ALCARAZ Marc (aka eKameleon) <ekameleon@gmail.com>.
  Portions created by the Initial Developer are Copyright (C) 2004-2011
  the Initial Developer. All Rights Reserved.
  
  Contributor(s) :
  
*/
package evideo.controllers.net 
{
    import core.maths.percentage;
    import core.strings.fastformat;

    import lunas.components.bars.LevelBar;

    import system.date.Time;

    import vegas.controllers.AbstractController;
    import vegas.net.NetStreamExpert;

    import flash.events.Event;
    import flash.text.TextField;
    
    /**
     * Invoked when the stream is in progress.
     */
    public class ProgressStream extends AbstractController 
    {
        /**
         * Creates a new ProgressStream instance.
         */
        public function ProgressStream()
        {
            super();
        }
        
        /**
         * The level bar of the video player.
         */
        public var bar:LevelBar ;
        
        /**
         * Indicates the expert reference of this display.
         */
        public var expert:NetStreamExpert ;
        
        /**
         * The video time textfield reference.
         */
        public var timeField:TextField ;
        
        /**
         * The video time pattern.
         */
        public var timePattern:String ;
        
        /**
         * Handles the event.
         */
        public override function handleEvent( e:Event ):void 
        {
            if ( expert == null )
            {
                logger.warn( this + " handleEvent failed with, the expert object not must be null or undefined." ) ;
                return ; 
            }
            var percent:Number = percentage( expert.netStream.bytesLoaded , expert.netStream.bytesTotal ) ;
            
            // logger.warn( this + " handleEvent :: percent:" + percent + " time:" + expert.time + " duration:" + expert.duration ) ;
            
            if ( _fixed )
            {
                return ;  
            }
            
            if ( bar )
            {
                bar.lock() ;
                bar.position = percent ;
                bar.level    = expert.time ;
                bar.unlock() ;
            }
            
            var current:String ;
            var time:Time = new Time() ; 
            var m:Number ;
            var s:Number ;
            
            if ( timeField )
            {
                time.setValue( expert.time , "s" ) ;
                
                m = time.getMinutes( 0 ) ;
                s = time.getSeconds( 0 ) ;
                
                current = "" ;
                
                if ( timePattern )
                {
                    current = fastformat
                    ( 
                        timePattern , 
                        (m <= 9) ? "0" + m :  m , 
                        (s <= 9) ? "0" + s :  s  
                    ) ;
                }
                
                timeField.htmlText = current ;
            }
        }
        
        /**
         * Invoked when start the dragging of the level bar.
         */
        public function startDragging( e:Event ):void
        {
            _fixed = true ;
        }
        
        /**
         * Invoked when stop the dragging of the level bar.
         */
        public function stopDragging( e:Event ):void
        {
            _fixed = false ;
        }
        
        /**
         * @private
         */
        private var _fixed:Boolean ;
    }
}
